

深度玩转Excel——第13周

DATAGURU专业数据分析社区

【声明】 本视频和幻灯片为炼数成金网络课程的教学资料，所有资料只能在课程内使用，不得在课程以外范围散播，违者将可能被追究法律和经济责任。

课程详情访问炼数成金培训网站

<http://edu.dataguru.cn>

关注炼数成金企业微信



■提供全面的数据价值资讯，涵盖商业智能与数据分析、大数据、企业信息化、数字化技术等，各种高性价比课程信息，赶紧掏出您的手机关注吧！



```
Sub WbAdd()  
    ' 程序创建 " 员工花名册 " 工作簿, 保存在本工作簿所在的文件夹中。  
    Dim Wb As Workbook, sht As Worksheet ' 定义一个 Workbook 对象和一个 Worksheet 对象  
    Set Wb = Workbooks.Add               ' 新建一个工作簿  
    Set sht = Wb.Worksheets(1)  
    With sht  
        .Name = " 花名册 "               ' 修改第一张工作表的标签名称  
        ' 设置表头  
        .Range("A1:F1") = Array("序号 ", "姓名 ", "性别 ", "出生年月 ", "参加工作时间 ", "备注 ")  
    End With  
    Wb.SaveAs ThisWorkbook.Path & "\ 员工花名册 .xls" ' 保存新建的工作表到本工作簿所在的文件夹中  
    ActiveWorkbook.Close                       ' 关闭新建的工作簿  
End Sub
```

判断工作簿是否打开

使用 Count 属性计算出当前打开的工作簿的个数。

```
Sub IsOpen()  
    '判断 "成绩表.xls" 工作簿文件是否打开。  
    Dim i As Integer  
    For i = 1 To Workbooks.Count  
        If Workbooks(i).Name = "成绩表.xls" Then  
            MsgBox "文件已打开！"  
            Exit Sub  
        End If  
    Next  
    MsgBox "文件没有打开！"  
End Sub
```

' 定义循环变量
' 开始循环
' 判断工作簿是否打开
' 如果找到该文件，退出过程

如果文件已打开，使用 Exit Sub 退出执行程序。

判断工作簿是否存在

```
Sub TestFile()  
    '判断本工作簿所在的文件夹中是否存在"员工花名册.xls"。  
    Dim fil As String '定义变量  
    fil = ThisWorkbook.Path & "\员工花名册.xls"  
    If Len(Dir(fil)) > 0 Then '用dir判断fil指代的文件是否存在  
        MsgBox "工作簿已存在!"  
    Else  
        MsgBox "工作簿不存在!"  
    End If  
End Sub
```

如果变量 fil 指代的文件存在，Dir 函数返回文件名，否则返回空字符串（""）。

向未打开的工作簿中录入数据

只有打开了工作簿才能进行编辑，所以先用代码打开它。

```
Sub WbInput()  
    '在本工作簿所在的文件夹下 "员工花名册" 里添加一条记录！  
    Dim wb As String, xrow As Integer, arr  
    wb = ThisWorkbook.Path & "\员工花名册.xls"           '指定要打开的文件  
    Workbooks.Open (wb)                                   '打开工作簿  
    With ActiveWorkbook.Worksheets(1)                     '向工作簿里的第1张工作表里添加记录  
        xrow = .Range("A1").CurrentRegion.Rows.Count + 1 '取得表格中第一条空行号  
        '将需要增加的职工信息保存在数组 arr 里  
        arr = Array(xrow - 1, "张姣", "女", #7/8/1987#, #9/1/2010#, "10年新招")  
        .Cells(xrow, 1).Resize(1, 6) = arr                '将数组写入单元格区域  
    End With  
    ActiveWorkbook.Close savechanges:=True                '关闭工作簿，并保存修改  
End Sub
```

```
Sub ShtAdd()  
    ' 根据 C 列的班级名新建不同的工作表。  
    Dim i As Integer, sht As Worksheet  
    i = 2  
    Set sht = Worksheets("成绩表")  
    Do While sht.Cells(i, "C") <> ""  
        Worksheets.Add after:=Worksheets(Worksheets.Count)  
        ActiveSheet.Name = sht.Cells(i, "C").Value  
        i = i + 1  
    Loop  
End Sub
```

' 第一条记录的行号为 2
' 定义循环条件
' 在所有工作表后插入新工作表
' 更改工作表的标签名称
' 行号增加 11


```
Sub FenLei()  
    '把成绩表按班级分到各个工作表中  
    Dim i As Long, bj As String, rng As Range  
    i = 2  
    bj = Cells(i, "C").Value  
    Do While bj <> ""  
        '将分表中A列第一个空单元格赋给 rng  
        Set rng = Worksheets(bj).Range("A65536").End(xlUp).Offset(1, 0)  
        Cells(i, "A").Resize(1, 7).Copy rng    '将记录复制到相应的工作表中  
        i = i + 1  
        bj = Cells(i, "C").Value  
    Loop  
End Sub
```

将工作表保存为新工作簿

使用 Mkdir 新建文件夹，变量 folder 指定文件夹的路径及名称。

```
Sub SaveToFile()  
    '把各个工作表以单独的工作簿文件保存在本工作簿所在文件夹下的 "班级成绩表" 文件夹中  
    Application.ScreenUpdating = False  
    Dim folder As String  
    folder = ThisWorkbook.Path & "\班级成绩表"  
    '如果文件夹不存在，新建文件夹  
    If Len(Dir(folder, vbDirectory)) = 0 Then Mkdir folder  
    Dim sht As Worksheet  
    For Each sht In Worksheets  
        sht.Copy  
        ActiveWorkbook.SaveAs folder & "\" & sht.Name & ".xls"  
        ActiveWorkbook.Close  
    Next  
    Application.ScreenUpdating = True  
End Sub
```

' 关闭屏幕更新

' 遍历工作表
' 复制工作表到新工作簿
' 保存工作簿，并命名

' 开启屏幕更新

```
Sub hebing()  
    ' 把各班成绩表合并到 " 总成绩 " 工作表中  
    Rows("2:65536").Clear  
    Dim sht As Worksheet, xrow As Integer, rng As Range  
    For Each sht In Worksheets  
        If sht.Name <> ActiveSheet.Name Then  
            Set rng = Range("A65536").End(xlUp).Offset(1, 0)  
            xrow = sht.Range("A1").CurrentRegion.Rows.Count - 1  
            sht.Range("A2").Resize(xrow, 7).Copy rng  
        End If  
    Next  
End Sub
```

' 删除原有记录

' 遍历工作簿中所有工作表

' 获得 A 列第一个空单元格

' 获得分表中的记录条数

' 粘贴记录到汇总表

汇总同文件夹下的多工作簿数据

代表待汇总的工作簿所在的文件夹路径的字符串，因为程序所在的工作簿与待汇总的工作簿在同一文件夹，所以写为“ThisWorkbook.Path”。

```
Sub HzwWb()  
    Dim bt As Range, r As Long, c As Long  
    r = 1          '1 是表头的行数  
    c = 8          '8 是表头的列数  
    Range(Cells(r + 1, "A"), Cells(65536, c)).ClearContents ' 清除汇总表中原表数据  
    Application.ScreenUpdating = False  
    Dim FileName As String, wb As Workbook, Erow As Long, fn As String, arr As Variant  
    FileName = Dir(ThisWorkbook.Path & "\*.xls")  
    Do While FileName <> ""  
        If FileName <> ThisWorkbook.Name Then ' 判断文件是否是本工作簿  
            Erow = Range("A1").CurrentRegion.Rows.Count + 1 ' 取得汇总表中第一条空行行号  
            fn = ThisWorkbook.Path & "\" & FileName  
            Set wb = GetObject(fn) ' 将 fn 代表的工作簿对象赋给变量  
            Set sht = wb.Worksheets(1) ' 汇总的是第 1 张工作表  
            ' 将数据表中的记录保存在 arr 数组里  
            arr = sht.Range(sht.Cells(r, "A"), sht.Cells(65536, "B").End(xlUp).Offset(0, 8))  
            ' 将数组 arr 中的数据写入工作表  
            Cells(Erow, "A").Resize(UBound(arr, 1), UBound(arr, 2)) = arr  
            wb.Close False  
        End If  
        FileName = Dir ' 用 Dir 函数取得其他文件名，并赋给变量  
    Loop  
    Application.ScreenUpdating = True  
End Sub
```

```
Sub mulu()  
    ' 为工作簿中所有工作表建立目录  
    Rows("2:65536").ClearContents  
    Dim sht As Worksheet, irow As Integer  
    irow = 2  
    For Each sht In Worksheets  
        Cells(irow, "A").Value = irow - 1  
        ' 写入工作表名, 并建立超链接  
        ActiveSheet.Hyperlinks.Add Anchor:=Cells(irow, "B"), Address:="",  
            SubAddress:="" & sht.Name & "!A1", TextToDisplay:=sht.Name  
        irow = irow + 1  
    Next  
End Sub
```

' 清除工作表中原有数据

' 在第 2 行写入第一条记录
' 遍历工作表
' 写入序号

' 行号加 1

向工作表中添加一个超链接对象 (Hyperlink)，其中 Anchor 指定建立超链接的位置，Address 参数指定超链接的地址，subAddress 参数为超链接的子地址，TextToDisplay 参数指定要显示的超链接的文本。


```
Private Sub Worksheet_Change(ByVal Target As Range)
' 如果更改的单元格不是 C 列第 3 行以下的单元格或更改的单元格个数大于 1 时退出程序
If Application.Intersect(Target, Range("C3:C65536")) Is Nothing Or Target.Count > 1 Then
Exit Sub
End If
Dim i As Integer      ' 定义变量
i = 3                  ' 参照表中第 1 条记录在第 3 行，所以初值设为 3
Do While Cells(i, "I").Value <> ""      ' 在参照表里循环
' 判断录入的字母与参照表的字母是否相符
If UCase(Target.Value) = Cells(i, "I").Value Then
Application.EnableEvents = False ' 禁用事件，防止将字母改为商品名称时，再次执行程序
Target.Value = Cells(i, "I").Offset(0, 1).Value      ' 写入产品名称
Target.Offset(0, -1).Value = Date                    ' 写入销售日期
Target.Offset(0, 1) = Cells(i, "I").Offset(0, 2).Value ' 写入商品代码
Target.Offset(0, 2) = Cells(i, "I").Offset(0, 3).Value ' 写入商品单价
Target.Offset(0, 3).Select                          ' 选中销售数量列，等待输入销售数量
Application.EnableEvents = True                      ' 重新启用事件
Exit Sub
End If
i = i + 1
Loop
End Sub
```

Intersect 方法返回参数指定的多个单元格的公共区域。参数至少是两个 Range 对象。

```
Private Sub Worksheet_SelectionChange(ByVal Target As Range)
    Range("B3:Q22").Interior.ColorIndex = xlNone ' 清除单元格里原有底纹颜色
    ' 当选中的单元格个数大于1时,重新给 Target 赋值
    If Target.Count > 1 Then
        Set Target = Target.Cells(1)
    End If
    ' 当选中的单元格不包含指定区域的单元格时,退出程序
    If Application.Intersect(Target, Range("B3:Q22")) Is Nothing Then
        Exit Sub
    End If
    Dim rng As Range ' 定义一个 Range 型变量
    ' 遍历单元格
    For Each rng In Range("B3:Q22")
        If rng.Value = Target.Value Then
            rng.Interior.ColorIndex = 39
        End If
    Next
End Sub
```

◆ 需求分析

- 问题描述
- 当前系统分析

◆ 功能分析

- 添加员工信息
- 删除员工信息
- 修改员工信息
- 查询员工信息
- 统计性别信息
- 统计工龄信息


◆ 表格设计

- 设计员工信息表
- 设计性别统计表
- 设计工龄统计表

◆ 设计窗体与代码

- 准备工作
- 设计添加员工信息窗体
- 设计删除员工信息窗体
- 设计修改员工信息窗体
- 设计查询员工信息窗体
- 设计人事管理系统主界面

实例：人事管理系统



◆ 需求分析

- 问题描述
- 分析当前系统

◆ 功能分析

- 解答填空题
- 解答判断题
- 解答单项选择题
- 解答多项选择题

◆ 设计卷面

- 设计填空题卷面
- 设计判断题卷面
- 设计单选题卷面
- 设计多选题卷面

◆ 设计窗体与代码

- 考试说明界面和分值说明界面
- 设计考试主界面



- ◆ Dataguru（炼数成金）是专业数据分析网站，提供教育，媒体，内容，社区，出版，数据分析业务等服务。我们的课程采用新兴的互联网教育形式，独创地发展了逆向收费式网络培训课程模式。既继承传统教育重学习氛围，重竞争压力的特点，同时又发挥互联网的威力打破时空限制，把天南地北志同道合的朋友组织在一起交流学习，使到原先孤立的学习个体组合成有组织的探索力量。并且把原先动辄成千上万的学习成本，直线下降至百元范围，造福大众。我们的目标是：低成本传播高价值知识，构架中国第一的网上知识流转阵地。
- ◆ 关于逆向收费式网络的详情，请看我们的培训网站 <http://edu.dataguru.cn>

Thanks

FAQ时间